

Termination Problem Data Base: format of input files

C. Marché A. Rubio H. Zantema

March 2, 2005

1 Syntax of TRS input files

1.1 grammar of input files

$$\begin{aligned} spec & ::= (decl) spec \mid \varepsilon \\ decl & ::= VAR idlist \mid THEORY listofthdecl \mid RULES listofrules \\ & \quad \mid STRATEGY strategydecl \mid id anylist \\ anylist & ::= \varepsilon \mid id anylist \mid string anylist \\ & \quad \mid (anylist) anylist \mid , anylist \\ idlist & ::= \varepsilon \mid id idlist \\ listofthdecl & ::= \varepsilon \mid (thdecl) listofthdecl \\ thdecl & ::= id idlist \mid EQUATIONS eqlist \\ eqlist & ::= \varepsilon \mid equation eqlist \\ equation & ::= term == term \\ listofrules & ::= \varepsilon \mid rule listofrules \\ rule & ::= term -> term \mid term -> term \mid condlist \\ & \quad \mid term ->= term \mid term ->= term \mid condlist \\ condlist & ::= cond \mid cond , condlist \\ cond & ::= term -> term \mid term -><- term \\ term & ::= id \mid id () \mid id (termlist) \\ termlist & ::= term \mid term , termlist \\ strategydecl & ::= INNERMOST \mid OUTERMOST \mid CONTEXTSENSITIVE csstratlist \\ csstratlist & ::= \varepsilon \mid (id intlist) csstratlist \\ intlist & ::= \varepsilon \mid int intlist \end{aligned}$$

id are non-empty sequences of characters except space, '(', ')', '"', and ', ' ; and excluding special sequences '->', '==', '->=', '-><-', '| ' and keywords

CONTEXTSENSITIVE, EQUATIONS, INNERMOST, OUTERMOST, RULES, STRATEGY, THEORY and VAR.

string are sequences of any characters between double quotes

int are non-empty sequences of digits

1.2 Semantical conditions

- at least one VAR and one RULES sections are mandatory. If they are several, the union is taken and :
 - a symbol declared in a VAR section must not have been used in previous declarations, and is assumed to denote a variable in remaining declarations (in particular must not be applied to arguments)
 - a symbol occurring in a RULES section which has not been used before is assumed to denote a function symbol, and must be used afterwards always with the same arity.

2 Syntax of SRS input files

2.1 grammar of input files

```
spec ::= ( decl ) spec | ε
decl ::= RULES listofrules | STRATEGY strategydecl | id anylist
anylist ::= ε | id anylist | string anylist | ( anylist ) anylist | , anylist
listofrules ::= ε | rule , listofrules
rule ::= word -> word | word ->
        | word ->= word | word ->=
word ::= id | id word
strategydecl ::= LEFTMOST | RIGHTMOST
```

id are non-empty sequences of characters except space, '(', ')', "'", and ', ' ; and excluding special sequences '->' and '->=', and keywords LEFTMOST, RIGHTMOST, RULES, and STRATEGY.

string are sequences of any characters between double quotes

2.2 Semantical conditions

- at least one RULES section is mandatory. If they are several, the union is taken.